/* STEP 1 */

/* *USE DIRECTEDGAR SOFTWARE TO SEACH FOR CONTARCTS IN EXHIBIT 10 (MATERIAL AGREEMENTS) ATTACHED TO 10K, 8K, AND 10Q SEC FILINGS;

*LIMIT THE SEARCH TO DOCUMENTS THAT INCLUDE WORDS "CREDIT AGREEMENT", "LOAN AGREEMENT", OR "LOAN AND SECURITY AGREEMENT"

*REQUIRE THE DOCUMENTS TO CONTAIN WORDS "FINANCIAL COVENANTS" and "GAAP" OR "GENERALLY ACCEPTED ACCOUNTING PRINCIPLES"; 

*SAVE DOCUMENTS IDENTIFIED BY THE SEARCH 

*USE DIRECTEDGAR TO RETAIN PARAGRAPS WITH (CONTEXT AROUND) EACH MENTION OF "GAAP" OR "GENERALLY ACCEPTED ACCOUNTING PRINCIPLES" IN EACH DOCUMENT 

*SAVE THE OUTPUT IN SEPARATE FILES */

/* STEP 2 */

/* Import the list of contracts to be used in analysis, the 
input files are based on DirectEdgar output */

/* import contracts identified from 8K filings */
proc import datafile='D:\Data\GAAP context\8K GAAP Contracts.txt'
     out=Contracts8K
     dbms=dlm
     replace;
     delimiter ='09'x ;
     datarow=3;
run;

/* read date/time variables */
Data contracts8k; 
set contracts8k; 
cik=scan(var1, 6, '\')*1; 
ryyyymmdd=scan(var1, 8, '\-'); 
cyyyymmdd=scan(var1, 9, '\-'); 
rmonth=substr(ryyyymmdd,6,2)*1;
ryear=substr(ryyyymmdd,2,4)*1; 
rday=substr(ryyyymmdd,8,2)*1;
cmonth=substr(cyyyymmdd,6,2)*1;
cyear=substr(cyyyymmdd,2,4)*1; 
cday=substr(cyyyymmdd,8,2)*1;
rdate=mdy(rmonth, rday, ryear);
cdate=mdy(cmonth, cday, cyear);
format rdate date9. cdate date9.;
path=var1;
source="08-K";
words=var5;
hits=var6;
relevance=var9;
size=var8;
run; 

data gaap.contracts8k; 
set contracts8k; 
keep cik rdate cdate path source words hits relevance size; run;

/* Import contracts identified from 10K filings */
proc import datafile='D:\Data\GAAP context\10K GAAP Contracts.txt'
     out=Contracts10K
     dbms=dlm
     replace;
     delimiter ='09'x ;
     datarow=3;
run;

/* Read date/time variables */
Data contracts10k; set contracts10k; 
cik=scan(var1, 6, '\')*1; 
ryyyymmdd=scan(var1, 8, '\-'); 
cyyyymmdd=scan(var1, 9, '\-'); 
rmonth=substr(ryyyymmdd,6,2)*1;
ryear=substr(ryyyymmdd,2,4)*1; 
rday=substr(ryyyymmdd,8,2)*1;
cmonth=substr(cyyyymmdd,6,2)*1;
cyear=substr(cyyyymmdd,2,4)*1; 
cday=substr(cyyyymmdd,8,2)*1;
rdate=mdy(rmonth, rday, ryear);
cdate=mdy(cmonth, cday, cyear);
format rdate date9. cdate date9.;
path=var1;
source="10-K";
words=var5;
hits=var6;
relevance=var9;
size=var8;
run; 

data gaap.contracts10k; set contracts10k; 
keep cik rdate cdate path source words hits relevance size; run;

/* Import contracts identified from 8Q filings */
proc import datafile='D:\Data\GAAP context\10Q GAAP Contracts.txt'
     out=Contracts10Q
     dbms=dlm
     replace;
     delimiter ='09'x ;
     datarow=3;
run;

/* Read date/time variables */
Data contracts10q; set contracts10q; 
cik=scan(var1, 6, '\')*1; 
ryyyymmdd=scan(var1, 8, '\-'); 
cyyyymmdd=scan(var1, 9, '\-'); 
rmonth=substr(ryyyymmdd,6,2)*1;
ryear=substr(ryyyymmdd,2,4)*1; 
rday=substr(ryyyymmdd,8,2)*1;
cmonth=substr(cyyyymmdd,6,2)*1;
cyear=substr(cyyyymmdd,2,4)*1; 
cday=substr(cyyyymmdd,8,2)*1;
rdate=mdy(rmonth, rday, ryear);
cdate=mdy(cmonth, cday, cyear);
format rdate date9. cdate date9.;
path=var1;
source="10-Q";
words=var5;
hits=var6;
relevance=var9;
size=var8;
run; 

data gaap.contracts10q; set contracts10q; 
keep cik rdate cdate path source words hits relevance size; run;

/* Stack 8K, 10K and 10Q contracts */

data gaap.contracts; 
set gaap.contracts10q gaap.contracts10k gaap.contracts8k; run;

/* Set filters: retain documents with over 7000 words to filter out false positives */
data gaap.contracts; 
set gaap.contracts; 
where words>7000 
and hits>15 and relevance>0.2; run;

/* this dataset has a list of contracts */
proc sort data=gaap.contracts nodupkey; by cik cdate rdate; run;


/* STEP 3 */

/* Input data: For each contract, use direct DirectEdgar software to retain paragraphs with (context around) each mention of GAAP */ 
/* Do this separately for contracts identified via 8Ks, 10Ks, or 10Qs */
/* The program below imports each input file and checks that each cell with context mentions GAAP */

proc import datafile="D:\Data\GAAP context\8K GAAP context part 1.csv"
     out=context8k1
     dbms=csv
     replace;
     getnames=yes;
run;

proc import datafile="D:\Data\GAAP context\8K GAAP context part 2.csv"
     out=context8k2
     dbms=csv
     replace;
     getnames=yes;
run;

data context8k; 
set context8k1 context8k2;
source="08-K";
; run; 

proc import datafile="D:\Data\GAAP context\10K GAAP context.csv"
     out=context10k
     dbms=csv
     replace;
     getnames=yes;
run;

data context10k; set context10k; 
source="10-K"; run;

proc import datafile="D:\Data\GAAP context\10Q GAAP context.csv"
     out=context10q
     dbms=csv
     replace;
     getnames=yes;
run;

data context10q; set context10q; 
source="10-Q"; run;

/* Stack context data */
data context_stacked; set context10k context10q context8k; 
run;

/* restrict to context mentions GAAP */
proc sql; create table context_stacked as select distinct
*
from context_stacked
where (upcase(lines) like "%GAAP%" or upcase(lines) like "%ACCOUNTING%PRINCIPLES%")=1
; QUIT;  


/* STEP 4 */

/* MERGE CONTEXT DATA FROM STEP 3 WITH THE LIST OF CONTRACTS FROM STEP 2*/

PROC SQL; CREATE TABLE CONTRACTS_CONTEXT AS SELECT DISTINCT
A.*, B.*
FROM GAAP.CONTRACTS A JOIN CONTEXT_STACKED B
ON A.CIK=B.CIK AND A.RDATE=B.RDATE AND A.CDATE=B.CDATE; QUIT;

DATA GAAP.CONTRACTS_CONTEXT; SET CONTRACTS_CONTEXT; RUN; 


***********************************************************************;

/* Step 5 */

/* USE DATA FROM STEP 4 TO CLASSIFY CONTRACTS INTO FROZEN/ROLING GAAP */

/* RUN IMPORTING NEW DATA FIRST */

proc sql; create table gaap.CONTRACT_CONTEXT_1 
as select distinct *, 
year(cdate) as cyear,
(lines like '%GAAP%') as GAAP,

max((lines like '%_ccounting _hange%'), 
(lines like '%GAAP _hange%')) as accounting_change,
(lines like '%amend%') as amend,
(lines like '%in effect from time to time%') as iefttt,
(lines like '%consistent%') as consistentbasis,

max( (lines like '%eliminate%effect%hange%'), (lines like '%in accordance with%prior to%hange%'),
(lines like '%in accordance with%before%hange%'), (lines like '%GAAP%in effect%before%hange%'),
(lines like '%rinciples%in effect%before%hange%'), (lines like '%GAAP%in effect%prior%hange%'),
(lines like '%rinciples%in effect%prior%hange%'), (lines like '%before the relevant%hange in%'),
(lines like '%in effect%immediately prior%hange%'), (lines like '%if%hange%had not occurred%'),
(lines like '%if%hange%had not been made%'), (lines like '%GAAP as in effect%the most recent%'),
(lines like '%rinciples%in effect%the most recent%'), (lines like '%no such change shall%'), 
(lines like '%calculations%basis consistent with%most recent%'), (lines like '%changes shall be followed%only from%date%'),
(lines like '%changes shall be followed%only after%date%'), (lines like '%calculations%made%in accordance with such change only after%') ) as excludechange,

max( (lines like '%negotiate in good faith%'), (lines like '%agree to enter%negotiations%'), 
(lines like '%agree to amend any such affected terms'),  (lines like '%request renegotiation%'), 
(lines like '%shall have amended'), (lines like '%equitably reflect%'),
(lines like '%reflect equitably%') ) as renegotiate,

max( (lines like '%GAAP%in effect on the _ate of this%'), (lines like '%rinciples%in effect on the _ate of this%'),
(lines like '%GAAP%in effect on the _ate hereof%'), (lines like '%rinciples%in effect on the _ate hereof%'),
(lines like '%GAAP%in effect on the _losing%'), (lines like '%rinciples%in effect on the _losing%'),
(lines like '%GAAP%in effect on the _ffective%'), (lines like '%rinciples%in effect on the _ffective%'),
(lines like '%GAAP%in effect on Jan%'), (lines like '%GAAP%in effect on Feb%'),
(lines like '%GAAP%in effect on Mar%'), (lines like '%GAAP%in effect on Apr%'),
(lines like '%GAAP%in effect on May%'), (lines like '%GAAP%in effect on Jun%'),
(lines like '%GAAP%in effect on Jul%'), (lines like '%GAAP%in effect on Aug%'),
(lines like '%GAAP%in effect on Sep%'), (lines like '%GAAP%in effect on Oct%'),
(lines like '%GAAP%in effect on Nov%'), (lines like '%GAAP%in effect on Dec%'),
(lines like '%rinciples%in effect on Jan%'), (lines like '%rinciples%in effect on Feb%'),
(lines like '%rinciples%in effect on Mar%'), (lines like '%rinciples%in effect on Apr%'),
(lines like '%rinciples%in effect on May%'), (lines like '%rinciples%in effect on Jun%'),
(lines like '%rinciples%in effect on Jul%'), (lines like '%rinciples%in effect on Aug%'),
(lines like '%rinciples%in effect on Sep%'), (lines like '%rinciples%in effect on Oct%'),
(lines like '%rinciples%in effect on Nov%'), (lines like '%rinciples%in effect on Dec%') ) as fixdate,

max( (lines like '%manner in which compliance%'),
(lines like '%compliance on such basis%'),
(lines like '%determining compliance%') ) as comply, 

max(calculated excludechange, calculated renegotiate, calculated fixdate) as fixed_gaap

from gaap.CONTRACTS_CONTEXT;
quit;


/* Aggregate the above classification by contract */
proc sql; create table gaap.gaap_panel as select
cik, cdate, max(year(cdate)) as cyear,
max(amend) as amend,
max(accounting_change) as accounting_change,

max(consistentbasis) as consistentbasis,
max(excludechange) as excludechange,
max(renegotiate) as renegotiate,
max(fixdate) as fixdate,
max(comply) as comply,
max(fixed_gaap) as fixed_gaap

from gaap.CONTRACT_CONTEXT_1
group by cik, cdate;
/*group by cik, cyear;*/
quit;


/* STEP 6 */ 

/* Download and prepare data from Compustat by running "XPF variables defined.sas" that generates the output file "xpf_controls" */

/* STEP 7 */

/* Link GAAP_Panel to Compustat *
/* File xpf.names is the Compustat table linking GVKEY to CIK */

proc sql; 
create table gaap_panel 
as select distinct a.*, b.gvkey, cik1
from gaap.gaap_panel a join xpf.names b 
on a.cik=b.cik1; 
quit;

proc sql; create table gaap.gaap_xpf as select distinct
a.*, b.*
from gaap_panel a join xpf.xpf_controls b 
on a.gvkey = b.gvkeyc and a.cyear=b.fyear
order by gvkey, fyear; quit;

/* Summary stats */

	proc sql; create table gaap.gaap_by_year_cst as select
	cyear, count(cik) as count,
	avg(fixed_gaap) as fixed_gaap

	from gaap.gaap_xpf
	where cyear>1993
	group by cyear
	order by cyear; quit;

